Computergestuetzte Mathematik zur Analysis

Lektion 11 (9. Januar)

```
> restart:
```

Implizite Funktionen

```
> g := solve(F(x, y), y);
```

$$g := RootOf(F(x, \_Z)) \qquad (1.1)$$

```
> diff(g, x);
```

$$-\frac{D_1(F)(x, RootOf(F(x, \_Z)))}{D_2(F)(x, RootOf(F(x, \_Z)))} \qquad (1.2)$$

```
> diff(g, x$2);
```

$$-\frac{1}{D_2(F)(x, RootOf(F(x, \_Z)))}\left(D_{1,1}(F)(x, RootOf(F(x, \_Z))) - \frac{D_{1,2}(F)(x, RootOf(F(x, \_Z)))\, D_1(F)(x, RootOf(F(x, \_Z)))}{D_2(F)(x, RootOf(F(x, \_Z)))}\right) + \frac{1}{D_2(F)(x, RootOf(F(x, \_Z)))^2}\left(D_1(F)(x, RootOf(F(x, \_Z)))\left(D_{1,2}(F)(x, RootOf(F(x, \_Z))) - \frac{D_{2,2}(F)(x, RootOf(F(x, \_Z)))\, D_1(F)(x, RootOf(F(x, \_Z)))}{D_2(F)(x, RootOf(F(x, \_Z)))}\right)\right) \qquad (1.3)$$

```
> alias(alpha = g);
```

$$\alpha \qquad (1.4)$$

```
> diff(alpha, x);
```

$$-\frac{D_1(F)(x, \alpha)}{D_2(F)(x, \alpha)} \qquad (1.5)$$

```
> normal(diff(alpha, x$2));
```

$$-\frac{1}{D_2(F)(x, \alpha)^3}\left(D_{1,1}(F)(x, \alpha)\, D_2(F)(x, \alpha)^2 - 2\, D_2(F)(x, \alpha)\, D_{1,2}(F)(x, \alpha)\, D_1(F)(x, \alpha) + D_{2,2}(F)(x, \alpha)\, D_1(F)(x, \alpha)^2\right) \qquad (1.6)$$

Das gleiche zu Fuss:

```
> diff(F(x, y(x)), x);
```

$$D_1(F)(x, y(x)) + D_2(F)(x, y(x))\left(\frac{d}{dx}\, y(x)\right) \quad (1.7)$$

```
> isolate((1.7)=0, diff(y(x),x)); #Aufloesen nach d/dx y(x)
```

$$\frac{d}{dx}\, y(x) = -\frac{D_1(F)(x, y(x))}{D_2(F)(x, y(x))} \quad (1.8)$$

```
> diff(F(x,y(x)),x$2);
```

$$D_{1,1}(F)(x, y(x)) + D_{1,2}(F)(x, y(x))\left(\frac{d}{dx}\, y(x)\right) + \left(D_{1,2}(F)(x, y(x)) + D_{2,2}(F)(x, y(x))\left(\frac{d}{dx}\, y(x)\right)\right)\left(\frac{d}{dx}\, y(x)\right) + D_2(F)(x, y(x))\left(\frac{d^2}{dx^2}\, y(x)\right) \quad (1.9)$$

```
> isolate(%=0,diff(y(x),x$2));
```

$$\frac{d^2}{dx^2}\, y(x) = \frac{1}{D_2(F)(x, y(x))}\left(-D_{1,1}(F)(x, y(x)) - D_{1,2}(F)(x, y(x))\left(\frac{d}{dx}\, y(x)\right) - \left(D_{1,2}(F)(x, y(x)) + D_{2,2}(F)(x, y(x))\left(\frac{d}{dx}\, y(x)\right)\right)\left(\frac{d}{dx}\, y(x)\right)\right) \quad (1.10)$$

```
> normal(algsubs((1.8),rhs((1.10)))); #Ersetze d/dx y(x) (1.8) in
  der RHS von (1.10)
```

$$-\frac{1}{D_2(F)(x, y(x))^3}\left(-2\, D_{1,2}(F)(x, y(x))\, D_1(F)(x, y(x))\, D_2(F)(x, y(x)) + D_{2,2}(F)(x, y(x))\, D_1(F)(x, y(x))^2 + D_{1,1}(F)(x, y(x))\, D_2(F)(x, y(x))^2\right) \quad (1.11)$$

Taylor-Entwicklung in mehreren Veraenderlichen

```
> f := (1-y^2)*exp(-x^2-y);
```

$$f := (1 - y^2)\, e^{-x^2 - y} \quad (2.1)$$

```
> mtaylor(f, [x=0, y=0], 4);
```

$$1 - y - x^2 - \frac{1}{2}\, y^2 + y\, x^2 + \frac{5}{6}\, y^3 \quad (2.2)$$

```
> for n from 1 to 9 do;
>   p[n] := mtaylor(f, [x=0, y=0], n);
> od:
> for n from 1 to 6 do;
>     'n' = n, 'p' = p[n];
> od;
```

$$n = 1,\ p = 1$$

$$n = 2,\ p = 1 - y$$

$$n = 3,\ p = 1 - y - x^2 - \frac{1}{2}\, y^2$$

$$n = 4,\ p = 1 - y - x^2 - \frac{1}{2}\,y^2 + y\,x^2 + \frac{5}{6}\,y^3$$

$$n = 5,\ p = 1 - y - x^2 - \frac{1}{2}\,y^2 + y\,x^2 + \frac{5}{6}\,y^3 + \frac{1}{2}\,x^4 + \frac{1}{2}\,x^2\,y^2 - \frac{11}{24}\,y^4$$

$$n = 6,\ p = 1 - y - x^2 - \frac{1}{2}\,y^2 + y\,x^2 + \frac{5}{6}\,y^3 + \frac{1}{2}\,x^4 + \frac{1}{2}\,x^2\,y^2 - \frac{11}{24}\,y^4 - \frac{1}{2}\,x^4\,y - \frac{5}{6}\,x^2\,y^3 + \frac{19}{120}\,y^5 \qquad \textbf{(2.3)}$$

```
> optionen := x = -1 .. 1, y = -1 .. 1, view = 0 .. 1.5,
  orientation = [-30, 70], axes = boxed, style = patchcontour,
  shading = zhue;
```

$$optionen := x = -1..1,\ y = -1..1,\ view = 0..1.5,\ orientation = [-30, 70],\ axes = boxed,\ style = patchcontour,\ shading = zhue \qquad \textbf{(2.4)}$$

```
> plot3d(f, optionen, title = "f");
```




```
> plot3d(p[3], optionen, title = "p[3]");
```

p[3]




```
> plot3d(p[6], optionen, title = "p[6]");
```

p[6]




```
> plot3d(p[9], optionen, title = "p[9]");
```

p[9]



Noch etwas Datenstrukturen

```
> liste1 := [28, 2^6, 0, -3];
```

$$liste1 := [28, 64, 0, -3] \quad (3.1)$$

```
> is_pos := t -> evalb(t > 0);
```

$$is\_pos := t \rightarrow evalb(0 < t) \quad (3.2)$$

```
> map(is_pos, liste1);
```

$$[true, true, false, false] \quad (3.3)$$

```
> select(is_pos, liste1);
```

$$[28, 64] \quad (3.4)$$

```
> f := cos(y^2);
```

$$f := \cos(y^2) \quad (3.5)$$

```
> has(f, y);
```

$$true \quad (3.6)$$

```
> has(f, y^2);
```

$$true \quad (3.7)$$

```
> has(f, y^3);
```

$$false \quad (3.8)$$

```
> liste := [x^3, exp(-4), x^2 + 15];
```

$$liste := [x^3, e^{-4}, x^2 + 15] \quad (3.9)$$

```
> test1 := t -> has(t, x^3);
```

$$test1 := t \rightarrow has(t, x^3) \quad (3.10)$$

```
> select(test1, liste);
```

$$[x^3] \quad (3.11)$$

```
> test2 := t -> has(t, exp);
```

$$test2 := t \rightarrow has(t, \exp) \quad (3.12)$$

```
> select(test2, liste);
```

$$[e^{-4}] \quad (3.13)$$

```
> test3 := t -> has(t, 15);
```

$$test3 := t \rightarrow has(t, 15) \quad (3.14)$$

```
> select(test3, liste);
```

$$[x^2 + 15] \quad (3.15)$$

```
> A := [seq(a[j], j = 1 .. 5)];
```

$$A := [a_1, a_2, a_3, a_4, a_5] \quad (3.16)$$

```
> B := [seq(b[j], j = 1 .. 4)];
```

$$B := [b_1, b_2, b_3, b_4] \quad (3.17)$$

```
> zip(F, A, B);
```

$$[F(a_1, b_1), F(a_2, b_2), F(a_3, b_3), F(a_4, b_4)] \quad (3.18)$$

```
> G := (a,b) -> (a,b);
```

$$G := (a, b) \rightarrow (a, b) \quad (3.19)$$

```
> zip(G, A, B);
```

$$[a_1, b_1, a_2, b_2, a_3, b_3, a_4, b_4] \quad (3.20)$$

```
> zip(G, 1, B);
```

$$[1, b_1, 1, b_2, 1, b_3, 1, b_4] \quad (3.21)$$